


# PRINT



This chapter covers the following topics:

- Function
- Syntax Description
- Example

For an explanation of the symbols used in the syntax diagram, see Syntax Symbols.

Related Statements: AT END OF PAGE | AT TOP OF PAGE | CLOSE PRINTER | DEFINE PRINTER | DISPLAY | EJECT | FORMAT | NEWPAGE | SKIP | SUSPEND IDENTICAL SUPPRESS | WRITE | WRITE TITLE | WRITE TRAILER

Belongs to Function Group: *Creation of Output Reports*

---

## Function

The PRINT statement is used to produce output in free format.

The PRINT statement differs from the WRITE statement in the following aspects:

- The output for each operand is written according to the value content rather than the length of the operand. Leading zeros for numeric values and trailing blanks for alphanumeric values are suppressed. The session parameter AD defines whether numeric values are printed left or right justified. With AD=L, the trailing blanks of a numeric value are suppressed. With AD=R, the leading blanks of a numeric value are printed.
- If the resulting output exceeds the current line size (LS parameter), the output is continued on the next line as follows: An alphanumeric constant or the content of an alphanumeric variable (without edit mask) is split at the rightmost blank or character which is neither a letter nor a numeric character contained on the current line. The first part of the split value is output to the current line, and the second part is written to the next line. Leading blanks in the second part are removed. As a consequence, empty lines are suppressed.

For all other operands, the entire value is written to the next line.

# Syntax Description

Operand Definition Table:

Operand	Possible Structure	Possible Formats	Referencing Permitted	Dynamic Definition
<i>operand1</i>	S A G N	A U N P I F B D T L G O	yes	no

Syntax Element Description:

Syntax Element	Description
<i>(rep)</i>	<p><b>Report Specification:</b></p> <p>The notation <i>(rep)</i> may be used to specify the identification of the report for which the PRINT statement is applicable.</p> <p>A value in the range 0 - 31 or a logical name which has been assigned using the DEFINE PRINTER statement may be specified.</p> <p>If <i>(rep)</i> is not specified, the PRINT statement will apply to the first report (Report 0).</p> <p>If this printer file is defined to Natural as PC, the report will be downloaded to the PC, see <i>Example 2</i>.</p> <p>For information on how to control the format of an output report created with Natural, see <i>Controlling Data Output</i> (in the <i>Programming Guide</i>).</p>

Syntax Element	Description
NOTITLE	<p><b>Default Page Title Suppression:</b></p> <p>Natural generates a single title line for each page resulting from a PRINT statement. This title contains the page number, the time of day, and the date. Time of day is set at the beginning of the session (TP mode) or at the beginning of the job (batch mode). This default title line may be overridden by using a WRITE TITLE statement, or it may be suppressed by specifying the NOTITLE clause in the PRINT statement. Examples:</p> <ul style="list-style-type: none"> <li>● Default title will be produced: <code>PRINT NAME</code></li> <li>● User title will be produced: <code>PRINT NAME WRITE TITLE 'user-title'</code></li> <li>● No title will be produced: <code>PRINT NOTITLE NAME</code></li> </ul> <p>If the NOTITLE option is used, it applies to all DISPLAY, PRINT and WRITE statements within the same object which write data to the same report.</p>
NOHDR	<p><b>Column Header Suppression:</b></p> <p>The PRINT statement itself does not produce any column headers. However, if you use the PRINT statement in conjunction with a DISPLAY statement, you can use the NOHDR option of the PRINT statement to suppress the column headers generated by the DISPLAY statement. The NOHDR option only takes effect if the execution of the PRINT statement causes a new page to be output.</p> <p>Without the NOHDR option, the column headers (if any) of the DISPLAY statement would be output on this new page; with NOHDR they will not.</p>

Syntax Element	Description
<i>statement-parameters</i>	<p><b>Parameter Definition at Statement Level:</b></p> <p>One or more parameters, enclosed within parentheses, may be specified at statement level, that is, immediately after the PRINT statement or an element being displayed.</p> <p>Each parameter specified in this manner will override any previous parameter specified in a GLOBALS command, SET GLOBALS (in Reporting Mode only) or FORMAT statement. If more than one parameter is specified, the parameters must be separated from one another by one or more blanks. A parameter entry must not be split between two statement lines.</p> <p>The parameter settings applied here will only be regarded for variable fields, but they have no effect on text-constants. If you would like to set field attributes for a text-constant, they have to be set explicitly for this element, see <i>Parameter Definition at Element (Field) Level</i>.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>List of Parameters</i></li> <li>● <i>Example of Parameter Usage at Statement and Element (Field) Level</i></li> </ul>
<i>nX, nT, /</i>	<p><b>Field Positioning, Text, Attribute Assignment:</b></p> <p>See <i>Field Positioning, Text, Attribute Assignment</i> below.</p>

## List of Parameters

Parameters that can be specified with the PRINT statement		Specification (S = at statement level, E = at element level)
AD	Attribute Definition	SE
AL	Alphanumeric Length for Output	SE
CD	Color Definition	SE
CV	Control Variable	SE
DF	Date Format	SE
DL	Display Length for Output	SE
DY	Dynamic Attributes	SE
EM	Edit Mask	SE
EMU	Unicode Edit Mask	E
FL	Floating Point Mantissa Length	SE
MC	Multiple-Value Field Count	S
MP	Maximum Number of Pages of a Report	S
NL	Numeric Length for Output	SE
PC	Periodic Group Count	S
PM	Print Mode	SE
SG	Sign Position	SE
ZP	Zero Printing	SE

The individual session parameters are described in the *Parameter Reference*.

## Example of Parameter Usage at Statement and Element (Field) Level

### Field Positioning, Text, Attribute Assignment

$\left\{ \left[ \begin{array}{l} n\mathbf{X} \\ n\mathbf{T} \\ / \end{array} \right] \dots \left\{ \begin{array}{l} 'text' [(attributes)] \\ 'c' (n) [(attributes)] \\ [='] operand1 [(parameters)] \end{array} \right\} \right\} \dots$
--

### Field Positioning Notations


Syntax Element	Description
<i>nX</i>	<p><b>Column Spacing:</b> This notation inserts <i>n</i> spaces between columns.</p> <p>PRINT NAME 5X SALARY</p>
<i>nT</i>	<p><b>Tab Setting:</b></p> <p>The <i>nT</i> notation causes positioning (tabulation) to print position <i>n</i>. Backward positioning results in a line advance.</p> <p>In the following example, NAME is printed beginning in position 25, and SALARY is printed beginning in position 50:</p> <p>PRINT 25T NAME 50T SALARY</p>
/	<p><b>Line Advance - Slash Notation:</b></p> <p>When placed between fields or text elements, a slash (/) causes positioning to the beginning of the next print line.</p> <p>PRINT NAME / SALARY</p>

### Text/Attribute Assignment

Syntax Element	Description
<code>'text'</code>	<p><b>Text Assignment:</b></p> <p>The character string enclosed by single quotes is displayed.</p> <pre>PRINT 'EMPLOYEE' NAME 'MARITAL/STATUS' MAR-STAT</pre>
<code>'c' (n)</code>	<p><b>Character Repetition:</b></p> <p>The character <i>c</i> enclosed by single quotes is displayed <i>n</i> times immediately before the field value.</p> <pre>PRINT '*' (5) '=' NAME</pre>
<code>'='</code>	<p><b>Field Content Positioned behind Field Heading:</b></p> <p>When placed before a field, the equal sign '=' results in the display of the field heading (as defined in the DEFINE DATA statement or in the DDM) followed by the field contents.</p> <pre>PRINT '=' NAME</pre>
<code>operand1</code>	<p><b>Field to be Printed:</b></p> <p>As <i>operand1</i> you specify the field to be printed.</p>
<code>parameters</code>	<p><b>Parameter Definition at Element (Field) Level:</b></p> <p>One or more parameters (see table above), enclosed within parentheses, may be specified immediately after <i>operand1</i>.</p> <p>Each parameter specified in this manner will override any previous parameter specified at statement level or in a GLOBALS command, SET GLOBALS (in Reporting Mode only) or FORMAT statement.</p> <p>If more than one parameter is specified, one or more blanks must be placed between each entry. An entry must not be split between two statement lines.</p> <p>See also:</p> <ul style="list-style-type: none"> <li>● <i>Statement Parameters</i></li> <li>● <i>Example of Parameter Usage at Statement and Element (Field) Level</i></li> </ul>

## Output Attributes

*attributes* indicates the output attributes to be used for text display. Attributes may be:



For the possible session parameter values, refer to the corresponding sections in the *Parameter Reference*:

- *AD - Attribute Definition*, section *Field Representation*
- *CD - Color Definition*
- *PM - Print Mode*

**Note:**

The compiler actually accepts more than one attribute value for an output field. For example, you may specify: AD=BDI. In such a case, however, only the last value applies. In the given example, only the value **I** will become effective and the output field will be displayed intensified.

## Example

- Example 1 - PRINT Statement
- Example 2 - PRINT Statement with Report to be Downloaded to the PC

### Example 1 - PRINT Statement

```

** Example 'PRTEX1': PRINT
*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 NAME
  2 FIRST-NAME
  2 CITY
  2 JOB-TITLE
  2 ADDRESS-LINE (2)
END-DEFINE
*
LIMIT 1
READ EMPLOY-VIEW BY CITY
/*
WRITE NOTITLE 'EXAMPLE 1:'
      // 'RESULT OF WRITE STATEMENT:'
WRITE      /  NAME  ',' FIRST-NAME ':' JOB-TITLE '**' (30)
WRITE      / 'RESULT OF PRINT STATEMENT:'
PRINT      /  NAME  ',' FIRST-NAME ':' JOB-TITLE '**' (30)
/*
WRITE      // 'EXAMPLE 2:'
      // 'RESULT OF WRITE STATEMENT:'
WRITE      /  NAME 60X ADDRESS-LINE (1:2)
WRITE      / 'RESULT OF PRINT STATEMENT:'

```



```
PRINT      /  NAME 60X ADDRESS-LINE (1:2)
/*
END-READ
END
```

**Output of Program PRTXEX1:**

EXAMPLE 1:

RESULT OF WRITE STATEMENT:

```
SENKO          , WILLIE          : PROGRAMMER
*****
```

RESULT OF PRINT STATEMENT:

```
SENKO , WILLIE : PROGRAMMER *****
```

EXAMPLE 2:

RESULT OF WRITE STATEMENT:

```
SENKO
2200 COLUMBIA PIKE    #914
```

RESULT OF PRINT STATEMENT:

```
SENKO                                     2200 COLUMBIA
PIKE #914
```

**Example 2 - PRINT Statement with Report to be Downloaded to the PC**

```
** Example 'PCPIEX1': PRINT to PC
**
** NOTE: Example requires that Natural Connection is installed.
*****
DEFINE DATA LOCAL
01 PERS VIEW OF EMPLOYEES
   02 PERSONNEL-ID
   02 NAME
   02 CITY
END-DEFINE
*
FIND PERS WITH CITY = 'NEW YORK'          /* Data selection
   PRINT (7) 5T CITY 20T NAME 40T PERSONNEL-ID /* (7) designates
                                                /* the output file
                                                /* (here the PC).

END-FIND
END
```